// File: @openzeppelin/contracts-upgradeable/access/OwnableUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/ContextUpgradeable.sol";
import "../proxy/utils/Initializable.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract OwnableUpgradeable is Initializable, ContextUpgradeable {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    function __Ownable_init() internal onlyInitializing {
        __Ownable_init_unchained();
    }

    function __Ownable_init_unchained() internal onlyInitializing {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[49] private __gap;
}

// File: @openzeppelin/contracts-upgradeable/proxy/utils/Initializable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (proxy/utils/Initializable.sol)

pragma solidity ^0.8.2;

import "../../utils/AddressUpgradeable.sol";

/**
 * @dev This is a base contract to aid in writing upgradeable contracts, or any kind of contract that will be deployed
 * behind a proxy. Since proxied contracts do not make use of a constructor, it's common to move constructor logic to an
 * external initializer function, usually called `initialize`. It then becomes necessary to protect this initializer
 * function so it can only be called once. The {initializer} modifier provided by this contract will have this effect.
 *
 * The initialization functions use a version number. Once a version number is used, it is consumed and cannot be
 * reused. This mechanism prevents re-execution of each "step" but allows the creation of new initialization steps in
 * case an upgrade adds a module that needs to be initialized.
 *
 * For example:
 *
 * [.hljs-theme-light.nopadding]
 * ```
 * contract MyToken is ERC20Upgradeable {
 *     function initialize() initializer public {
 *         __ERC20_init("MyToken", "MTK");
 *     }
 * }
 * contract MyTokenV2 is MyToken, ERC20PermitUpgradeable {
 *     function initializeV2() reinitializer(2) public {
 *         __ERC20Permit_init("MyToken");
 *     }
 * }
 * ```
 *
 * TIP: To avoid leaving the proxy in an uninitialized state, the initializer function should be called as early as
 * possible by providing the encoded function call as the `_data` argument to {ERC1967Proxy-constructor}.
 *
 * CAUTION: When used with inheritance, manual care must be taken to not invoke a parent initializer twice, or to ensure
 * that all initializers are idempotent. This is not verified automatically as constructors are by Solidity.
 *
 * [CAUTION]
 * ====
 * Avoid leaving a contract uninitialized.
 *
 * An uninitialized contract can be taken over by an attacker. This applies to both a proxy and its implementation
 * contract, which may impact the proxy. To prevent the implementation contract from being used, you should invoke
 * the {_disableInitializers} function in the constructor to automatically lock it when it is deployed:
 *
 * [.hljs-theme-light.nopadding]
 * ```
 * /// @custom:oz-upgrades-unsafe-allow constructor
 * constructor() {
 *     _disableInitializers();
 * }
 * ```
 * ====
 */
abstract contract Initializable {
    /**
     * @dev Indicates that the contract has been initialized.
     * @custom:oz-retyped-from bool
     */
    uint8 private _initialized;

    /**
     * @dev Indicates that the contract is in the process of being initialized.
     */
    bool private _initializing;

    /**
     * @dev Triggered when the contract has been initialized or reinitialized.
     */
    event Initialized(uint8 version);

    /**
     * @dev A modifier that defines a protected initializer function that can be invoked at most once. In its scope,
     * `onlyInitializing` functions can be used to initialize parent contracts.
     *
     * Similar to `reinitializer(1)`, except that functions marked with `initializer` can be nested in the context of a
     * constructor.
     *
     * Emits an {Initialized} event.
     */
    modifier initializer() {
        bool isTopLevelCall = !_initializing;
        require(
            (isTopLevelCall && _initialized < 1) || (!AddressUpgradeable.isContract(address(this)) && _initialized == 1),
            "Initializable: contract is already initialized"
        );
        _initialized = 1;
        if (isTopLevelCall) {
            _initializing = true;
        }
        _;
        if (isTopLevelCall) {
            _initializing = false;
            emit Initialized(1);
        }
    }

    /**
     * @dev A modifier that defines a protected reinitializer function that can be invoked at most once, and only if the
     * contract hasn't been initialized to a greater version before. In its scope, `onlyInitializing` functions can be
     * used to initialize parent contracts.
     *
     * A reinitializer may be used after the original initialization step. This is essential to configure modules that
     * are added through upgrades and that require initialization.
     *
     * When `version` is 1, this modifier is similar to `initializer`, except that functions marked with `reinitializer`
     * cannot be nested. If one is invoked in the context of another, execution will revert.
     *
     * Note that versions can jump in increments greater than 1; this implies that if multiple reinitializers coexist in
     * a contract, executing them in the right order is up to the developer or operator.
     *
     * WARNING: setting the version to 255 will prevent any future reinitialization.
     *
     * Emits an {Initialized} event.
     */
    modifier reinitializer(uint8 version) {
        require(!_initializing && _initialized < version, "Initializable: contract is already initialized");
        _initialized = version;
        _initializing = true;
        _;
        _initializing = false;
        emit Initialized(version);
    }

    /**
     * @dev Modifier to protect an initialization function so that it can only be invoked by functions with the
     * {initializer} and {reinitializer} modifiers, directly or indirectly.
     */
    modifier onlyInitializing() {
        require(_initializing, "Initializable: contract is not initializing");
        _;
    }

    /**
     * @dev Locks the contract, preventing any future reinitialization. This cannot be part of an initializer call.
     * Calling this in the constructor of a contract will prevent that contract from being initialized or reinitialized
     * to any version. It is recommended to use this to lock implementation contracts that are designed to be called
     * through proxies.
     *
     * Emits an {Initialized} event the first time it is successfully executed.
     */
    function _disableInitializers() internal virtual {
        require(!_initializing, "Initializable: contract is initializing");
        if (_initialized < type(uint8).max) {
            _initialized = type(uint8).max;
            emit Initialized(type(uint8).max);
        }
    }

    /**
     * @dev Internal function that returns the initialized version. Returns `_initialized`
     */
    function _getInitializedVersion() internal view returns (uint8) {
        return _initialized;
    }

    /**
     * @dev Internal function that returns the initialized version. Returns `_initializing`
     */
    function _isInitializing() internal view returns (bool) {
        return _initializing;
    }
}

// File: @openzeppelin/contracts-upgradeable/utils/AddressUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library AddressUpgradeable {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts-upgradeable/utils/ContextUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;
import "../proxy/utils/Initializable.sol";

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract ContextUpgradeable is Initializable {
    function __Context_init() internal onlyInitializing {
    }

    function __Context_init_unchained() internal onlyInitializing {
    }
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }

    /**
     * @dev This empty reserved space is put in place to allow future versions to add new
     * variables without shifting down storage in the inheritance chain.
     * See https://docs.openzeppelin.com/contracts/4.x/upgradeable#storage_gaps
     */
    uint256[50] private __gap;
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: contracts/interface/IStakeFactory.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity ^0.8.0;

interface IStakeFactory {
    function impl() external view returns(address);
}
// File: contracts/interface/IStaking3.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity ^0.8.0;

interface IStaking3 {
    function deposit(address _user, uint128 _amount, uint128 _stakeType) external payable returns(uint128,uint128, uint128);
    function compound(address _user, uint128 _stakeType) external payable returns(uint128,uint128, uint128);
    function withdraw(address _user, uint128 _amount, uint128 _stakeType) external payable returns(uint128,uint128, uint128);
    function claim(address _user, uint128 _stakeType) external payable returns(uint128,uint128);
    function addStakedType(uint128 _stakePeriod, uint128 _depositFees, uint128 _withdrawalFees, uint128 _rewardRate) external returns(uint128);
    function updateStakeType(uint128 _stakeType, uint128 _stakePeriod, uint128 _depositFees, uint128 _withdrawalFees, uint128 _rewardRate) external;
    function deleteStakeType(uint128 _stakeType) external returns(bool);
    function emergencyWithdraw(address _user, uint128 _stakeType) external payable returns(uint128,uint128,uint128);
    function updateEmergencyFees(uint128 newFee) external ;
    function updatePlatformFee(uint128 newFee) external;
    function updateOwnerWallet(address newOwnerWallet) external;
    function updateTreasuryWallet(address newTreasurywallet) external;
    function updateStakeLimit(uint128 _newLimit) external;
    function getPoolLength() external view returns(uint128);
}   
// File: contracts/interface/IToken.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity ^0.8.0;

interface IToken {
    function name() external view returns(string memory);
        function symbol() external view returns(string memory);

}
// File: contracts/SaitaProxy.sol



// SPDX-License-Identifier: Unlicensed

pragma solidity ^0.8.0;

import "./interface/IStakeFactory.sol";

contract SaitaProxy {
    
        bytes32 private constant proxyOwnerPosition = keccak256("com.saitama.proxy.owner");
        bytes32 private constant factory = keccak256("com.saitama.proxy.factory");

    constructor(address owner) {
        setProxyOwner(owner);

    }

    function setProxyOwner(address newProxyOwner) private  {
        bytes32 position = proxyOwnerPosition;
        assembly {
            sstore(position, newProxyOwner)
        }
    }

    function setFactory(address _factory) public  {
        require(msg.sender == proxyOwner(), "ONLY_OWNER_CAN_CHANGE");
        bytes32 position = factory;
        assembly {
            sstore(position, _factory)
        }
    }

    function getFactory() public view returns (address _factory) {
        bytes32 position = factory;
        assembly {
            _factory := sload(position)
        }
    }

    function proxyOwner() public view returns (address owner) {
        bytes32 position = proxyOwnerPosition;
        assembly {
            owner := sload(position)
        }
    }


    function implementation() public view returns (address) {
        return IStakeFactory(getFactory()).impl();
    }
    


    fallback() external payable {
        address _impl = implementation();

            assembly 
                {
                let ptr := mload(0x40)

                // (1) copy incoming call data
                calldatacopy(ptr, 0, calldatasize())

                // (2) forward call to logic contract
                let result := delegatecall(gas(), _impl, ptr, calldatasize(), 0, 0)
                let size := returndatasize()

                // (3) retrieve return data
                returndatacopy(ptr, 0, size)

                // (4) forward return data back to caller
                switch result
                case 0 { revert(ptr, size) }
                default { return(ptr, size) }
                }
    }

}
// File: contracts/SaitaStakingFactory.sol



// SPDX-License-Identifier: UNLICENSED

pragma solidity ^0.8.0;

import "@openzeppelin/contracts-upgradeable/access/OwnableUpgradeable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./SaitaProxy.sol";
import "./interface/IStaking3.sol";
import "./interface/IToken.sol";


contract SaitaStakingFactory is OwnableUpgradeable{

    address internal imp;
    address[] public totalStakingInstances;

    mapping(address => address) public tokenAddrToStakingAddr;

    event Deposit(address indexed stakeProxy, uint128 stakeAmount, uint128 stakeType, address indexed user, uint256 blockTimestamp, uint128 period, uint128 totalStakedInPool);
    event Compound(address indexed stakeProxy, uint128 amount, uint128 stakeType, address indexed user, uint256 blockTimestamp, uint128 period, uint128 totalStakedInPool);
    event Withdraw(address indexed stakeProxy, uint128 amount, uint128 stakeType, address indexed user, uint256 blockTimestamp, uint128 period, uint128 totalStakedInPool);
    event Claim(address indexed stakeProxy,  uint128 amount, uint128 stakeType, address indexed user, uint256 blockTimestamp, uint128 period);
    event AddStakeType(address indexed stakeProxy, uint128 stakeType, uint128 stakePeriod, uint128 depositFees, uint128 withdrawalFees, uint128 rewardRate, address indexed caller, uint256 blockTimestamp);
    event UpdateStakeType(address indexed stakeProxy, uint128 stakeType, uint128 stakePeriod, uint128 depositFees, uint128 withdrawalFees, uint128 rewardRate, address indexed caller, uint256 blockTimestamp);
    event DeleteStakeType(address indexed stakeProxy, uint128 stakeType, address indexed caller, uint256 blockTimestamp, bool active);
    event EmergencyWithdrawn(address indexed stakeProxy,  uint128 amount, uint128 stakeType, address indexed user, uint256 blockTimestamp, uint128 period, uint128 totalStakedInPool);
    event UpdateEmergencyFees(address indexed _owner, uint128 newFees, uint256 time);
    event UpdatePlatformFee(address indexed stakeProxy, uint128 newFee);
    event UpdateOwnerWallet(address indexed stakeProxy, address indexed newOwnerWallet);
    event UpdateTreasuryWallet(address indexed stakeProxy, address indexed newTreasuryWallet);
    event UpdateStakeLimit(address indexed stakeProxy, uint128 _newLimit);

    event Deployed(address indexed instance);
    event StakingInitialized(address indexed proxyInstance, address indexed stakedToken, uint128 stakePeriod, uint128 depositFees, uint128 withdrawlsFees, uint128 rewardRate, string name, string _symbol, uint256 blockTimestamp, string uri);

    function initialize(address _imp) external initializer {
        __Ownable_init();
        imp = _imp;
    }

    function deposit(address _stakedToken, uint128 _stakeAmount, uint128 _stakeType) external payable {
        IERC20(_stakedToken).approve(tokenAddrToStakingAddr[_stakedToken], _stakeAmount);
        (uint128 amount, uint128 period, uint128 totalStakedInPool) = IStaking3(tokenAddrToStakingAddr[_stakedToken]).deposit{value:msg.value}(msg.sender, _stakeAmount, _stakeType);
        emit Deposit(tokenAddrToStakingAddr[_stakedToken], amount, _stakeType, msg.sender, block.timestamp, period, totalStakedInPool);
    }
    
    function compound(address _stakedToken, uint128 _stakeType) external payable {
        (uint128 amount, uint128 period, uint128 totalStakedInPool) = IStaking3(tokenAddrToStakingAddr[_stakedToken]).compound{value:msg.value}(msg.sender, _stakeType);
        emit Compound(tokenAddrToStakingAddr[_stakedToken], amount, _stakeType, msg.sender, block.timestamp, period, totalStakedInPool);
    }

    function withdraw(address _stakedToken, uint128 _amount, uint128 _stakeType) external payable {
        IERC20(_stakedToken).approve(tokenAddrToStakingAddr[_stakedToken], _amount);
        (uint128 amount, uint128 period, uint128 totalStakedInPool) = IStaking3(tokenAddrToStakingAddr[_stakedToken]).withdraw{value:msg.value}(msg.sender, _amount, _stakeType);
        emit Withdraw(tokenAddrToStakingAddr[_stakedToken], amount, _stakeType, msg.sender, block.timestamp, period, totalStakedInPool);
    }

    function claim(address _stakedToken, uint128 _stakeType) external payable {
        (uint128 amount, uint128 period) = IStaking3(tokenAddrToStakingAddr[_stakedToken]).claim{value:msg.value}(msg.sender, _stakeType);
        emit Claim(tokenAddrToStakingAddr[_stakedToken], amount, _stakeType, msg.sender, block.timestamp, period);
    }

    function emergencyWithdraw(address _stakedToken, uint128 _stakeType) external payable {
        (uint128 amount, uint128 period, uint128 totalStakedInPool) = IStaking3(tokenAddrToStakingAddr[_stakedToken]).emergencyWithdraw{value:msg.value}(msg.sender, _stakeType);
        emit EmergencyWithdrawn(tokenAddrToStakingAddr[_stakedToken], amount, _stakeType, msg.sender, block.timestamp, period, totalStakedInPool);
    }

    function addStakedType(address _stakedToken, uint128 _stakePeriod, uint128 _depositFees, uint128 _withdrawalFees, uint128 _rewardRate) external onlyOwner{
        uint128 _stakeType = IStaking3(tokenAddrToStakingAddr[_stakedToken]).addStakedType(_stakePeriod,_depositFees,_withdrawalFees,_rewardRate);
        emit AddStakeType(tokenAddrToStakingAddr[_stakedToken], _stakeType, _stakePeriod, _depositFees, _withdrawalFees, _rewardRate, msg.sender, block.timestamp);
    }

    function updateStakeType(address _stakedToken, uint128 _stakeType, uint128 _stakePeriod, uint128 _depositFees, uint128 _withdrawalFees, uint128 _rewardRate) external onlyOwner{
        IStaking3(tokenAddrToStakingAddr[_stakedToken]).updateStakeType(_stakeType,_stakePeriod,_depositFees,_withdrawalFees,_rewardRate);
        emit UpdateStakeType(tokenAddrToStakingAddr[_stakedToken], _stakeType, _stakePeriod, _depositFees, _withdrawalFees, _rewardRate,msg.sender, block.timestamp);
    }

    function deleteStakeType(address _stakedToken, uint128 _stakeType) external onlyOwner {
        bool active = IStaking3(tokenAddrToStakingAddr[_stakedToken]).deleteStakeType(_stakeType);
        emit DeleteStakeType(tokenAddrToStakingAddr[_stakedToken], _stakeType,msg.sender, block.timestamp, active);
    }
    function updateEmergencyFees(address _stakedToken, uint128 newFees) external onlyOwner {
        IStaking3(tokenAddrToStakingAddr[_stakedToken]).updateEmergencyFees(newFees);
        emit UpdateEmergencyFees(msg.sender, newFees, block.timestamp);
    }

    function updatePlatformFee(address _stakedToken, uint128 newFee) external onlyOwner {
        IStaking3(tokenAddrToStakingAddr[_stakedToken]).updatePlatformFee(newFee);

        emit UpdatePlatformFee(tokenAddrToStakingAddr[_stakedToken], newFee);
    }

    function updateOwnerWallet(address _stakedToken, address newOwnerWallet) external onlyOwner {
       IStaking3(tokenAddrToStakingAddr[_stakedToken]).updateOwnerWallet(newOwnerWallet);

        emit UpdateOwnerWallet(tokenAddrToStakingAddr[_stakedToken], newOwnerWallet);
    }

    function updateTreasuryWallet(address _stakedToken, address newTreasuryWallet) external onlyOwner {
        IStaking3(tokenAddrToStakingAddr[_stakedToken]).updateTreasuryWallet(newTreasuryWallet);

        emit UpdateTreasuryWallet(tokenAddrToStakingAddr[_stakedToken], newTreasuryWallet);
    }

    function getPoolLength(address _stakedToken) external view returns(uint128) {
        return IStaking3(tokenAddrToStakingAddr[_stakedToken]).getPoolLength();
    }

    function deployInstance() external onlyOwner returns(address addr) {
        addr = address(new SaitaProxy(address(this)));
        require(addr!=address(0), "NULL_CONTRACT_ADDRESS_CREATED");
        totalStakingInstances.push(addr);
        
        emit Deployed(addr);
        return addr;
    }

    function initializeProxyInstance(address _proxyInstance, 
                                    address _ownerWallet, 
                                    address _stakedToken, 
                                    uint128 _stakePeriod, 
                                    uint128 _depositFees, 
                                    uint128 _withdrawalFees, 
                                    uint128 _rewardRate, 
                                    uint128 _emergencyFees, 
                                    uint128 _platformFee, 
                                    address _treasury, 
                                    uint128 _maxStakeLimit, 
                                    string memory uri) external onlyOwner {
        
            tokenAddrToStakingAddr[_stakedToken] = _proxyInstance;
        {
            string memory _name = IToken(_stakedToken).name();
            string memory _symbol = IToken(_stakedToken).symbol();

            // initialize pool
            SaitaProxy proxyInstance = SaitaProxy(payable(_proxyInstance));
            {
                bytes memory init = returnHash(_ownerWallet, _stakedToken, _stakePeriod, _depositFees, _withdrawalFees, _rewardRate, _emergencyFees, _platformFee, _treasury, _maxStakeLimit);
                if (init.length > 0)
                    
                    assembly 
                    {
                        if eq(call(gas(), proxyInstance, 0, add(init, 0x20), mload(init), 0, 0), 0) {
                            revert(0, 0)
                        }
                    }
            }
            emit StakingInitialized(_proxyInstance, _stakedToken, _stakePeriod, _depositFees, _withdrawalFees, _rewardRate, _name, _symbol, block.timestamp, uri);
        }
    }

    function setFactoryProxyInstance(address _proxyInstance) external onlyOwner {
                // set factory in deployed proxy instance
                SaitaProxy proxyInstance = SaitaProxy(payable(_proxyInstance));
                proxyInstance.setFactory(address(this));
    }

    function returnHash(address _ownerWallet, address _stakedToken, 
                        uint128 _stakePeriod, uint128 _depositFees, 
                        uint128 _withdrawalFees, uint128 _rewardRate, uint128 _emergencyFees, uint128 _platformFee, address _treasury, uint128 _maxStakeLimit) internal pure returns(bytes memory data) {
        data = abi.encodeWithSignature("initialize(address,address,uint128,uint128,uint128,uint128,uint128,uint128,address,uint128)", _ownerWallet,_stakedToken,_stakePeriod,_depositFees,_withdrawalFees,_rewardRate, _emergencyFees,_platformFee,_treasury, _maxStakeLimit);
    }

    function updateImp(address _newImp) external onlyOwner {
        imp = _newImp;
    }

    function impl() external view returns(address) {
        return imp;
    }

    function totalStakingNo() external view returns(uint256) {
        return totalStakingInstances.length;
    }

    function updateStakeLimit(address _stakedToken, uint128 _newLimit) external onlyOwner {
        IStaking3(tokenAddrToStakingAddr[_stakedToken]).updateStakeLimit(_newLimit);
        emit UpdateStakeLimit(tokenAddrToStakingAddr[_stakedToken], _newLimit);
    }
    
}